

Mathematicians define a relation  $R$  to be a set of ordered pairs, and write  $s R t$  to mean  $\langle s, t \rangle \in R$ . The transitive closure  $TC(R)$  of the relation  $R$  is the smallest relation containing  $R$  such that,  $s TC(R)t$  and  $t TC(R)u$  imply  $s TC(R)u$ , for any  $s, t$ , and  $u$ . This module shows several ways of defining the operator  $TC$ .

It is sometimes more convenient to represent a relation as a Boolean-valued function of two arguments, where  $s R t$  means  $R[s, t]$ . It is a straightforward exercise to translate everything in this module to that representation.

Mathematicians say that  $R$  is a relation on a set  $S$  iff  $R$  is a subset of  $S \times S$ . Let the *support* of a relation  $R$  be the set of all elements  $s$  such that  $s R t$  or  $t R s$  for some  $t$ . Then any relation is a relation on its support. Moreover, the support of  $R$  is the support of  $TC(R)$ . So, to define the transitive closure of  $R$ , there's no need to say what set  $R$  is a relation on.

Let's begin by importing some modules we'll need and defining the support of a relation.

EXTENDS *Integers, Sequences, FiniteSets, TLC*

$$Support(R) \triangleq \{r[1] : r \in R\} \cup \{r[2] : r \in R\}$$

A relation  $R$  defines a directed graph on its support, where there is an edge from  $s$  to  $t$  iff  $s R t$ . We can define  $TC(R)$  to be the relation such that  $s R t$  holds iff there is a path from  $s$  to  $t$  in this graph. We represent a path by the sequence of nodes on the path, so the length of the path (the number of edges) is one greater than the length of the sequence. We then get the following definition of  $TC$ .

$$\begin{aligned} TC(R) &\triangleq \\ \text{LET } S &\triangleq Support(R) \\ \text{IN } \{ \langle s, t \rangle \in S \times S : & \\ \quad \exists p \in Seq(S) : \wedge Len(p) > 1 & \\ \quad \wedge p[1] = s & \\ \quad \wedge p[Len(p)] = t & \\ \quad \wedge \forall i \in 1 .. (Len(p) - 1) : \langle p[i], p[i+1] \rangle \in R \} & \end{aligned}$$

This definition can't be evaluated by *TLC* because  $Seq(S)$  is an infinite set. However, it's not hard to see that if  $R$  is a finite set, then it suffices to consider paths whose length is at most  $Cardinality(S)$ . Modifying the definition of  $TC$  we get the following definition that defines  $TC1(R)$  to be the transitive closure of  $R$ , if  $R$  is a finite set. The *LET* expression defines  $BoundedSeq(S, n)$  to be the set of all sequences in  $Seq(S)$  of length at most  $n$ .

$$\begin{aligned} TC1(R) &\triangleq \\ \text{LET } BoundedSeq(S, n) &\triangleq \text{UNION } \{ [1 .. i \rightarrow S] : i \in 0 .. n \} \\ S &\triangleq Support(R) \\ \text{IN } \{ \langle s, t \rangle \in S \times S : & \\ \quad \exists p \in BoundedSeq(S, Cardinality(S) + 1) : & \\ \quad \wedge Len(p) > 1 & \\ \quad \wedge p[1] = s & \\ \quad \wedge p[Len(p)] = t & \\ \quad \wedge \forall i \in 1 .. (Len(p) - 1) : \langle p[i], p[i+1] \rangle \in R \} & \end{aligned}$$

This naive method used by *TLC* to evaluate expressions makes this definition rather inefficient. (As an exercise, find an upper bound on its complexity.) To obtain a definition that *TLC* can evaluate more efficiently, let's look at the closure operation more algebraically. Let's define the composition of two relations  $R$  and  $T$  as follows.

$$\begin{aligned}
 R ** T &\triangleq \text{LET } SR \triangleq \text{Support}(R) \\
 &\quad ST \triangleq \text{Support}(T) \\
 &\quad \text{IN } \{ \langle r, t \rangle \in SR \times ST : \\
 &\quad \quad \exists s \in SR \cap ST : (\langle r, s \rangle \in R) \wedge (\langle s, t \rangle \in T) \}
 \end{aligned}$$

We can then define the closure of  $R$  to equal

$$R \cup (R ** R) \cup (R ** R ** R) \cup \dots$$

For  $R$  finite, this union converges to the transitive closure when the number of terms equals the cardinality of the support of  $R$ . This leads to the following definition.

$$\begin{aligned}
 TC2(R) &\triangleq \\
 \text{LET } C[n \in \text{Nat}] &\triangleq \text{IF } n = 0 \text{ THEN } R \\
 &\quad \text{ELSE } C[n - 1] \cup (C[n - 1] ** R) \\
 \text{IN } \text{IF } R = \{\} &\text{ THEN } \{\} \text{ ELSE } C[\text{Cardinality}(\text{Support}(R)) - 1]
 \end{aligned}$$

These definitions of *TC1* and *TC2* are somewhat unsatisfactory because of their use of *Cardinality*( $S$ ). For example, it would be easy to make a mistake and use *Cardinality*( $S$ ) instead of *Cardinality*( $S$ ) + 1 in the definition of *TC1*( $R$ ). I find the following definition more elegant than the preceding two. It is also more asymptotically more efficient because it makes  $O(\log \text{Cardinality}(S))$  rather than  $O(\text{Cardinality}(S))$  recursive calls.

$$\begin{aligned}
 \text{RECURSIVE } TC3(-) & \\
 TC3(R) &\triangleq \text{LET } RR \triangleq R ** R \\
 &\quad \text{IN } \text{IF } RR \subseteq R \text{ THEN } R \text{ ELSE } TC3(R \cup RR)
 \end{aligned}$$

The preceding two definitions can be made slightly more efficient to execute by expanding the definition of  $**$  and making some simple optimizations. But, this is unlikely to be worth complicating the definitions for.

The following definition is (asymptotically) the most efficient. It is essentially the TLA+ representation of *Warshall's algorithm*. (*Warshall's algorithm* is typically written as an iterative procedure for the case of a relation on a set  $i \dots j$  of integers, when the relation is represented as a Boolean-valued function.)

$$\begin{aligned}
 TC4(R) &\triangleq \\
 \text{LET } S &\triangleq \text{Support}(R) \\
 \text{RECURSIVE } TCR(-) & \\
 TCR(T) &\triangleq \text{IF } T = \{\} \\
 &\quad \text{THEN } R \\
 &\quad \text{ELSE LET } r \triangleq \text{CHOOSE } s \in T : \text{TRUE} \\
 &\quad \quad RR \triangleq TCR(T \setminus \{r\}) \\
 &\quad \quad \text{IN } RR \cup \{ \langle s, t \rangle \in S \times S : \\
 &\quad \quad \quad \langle s, r \rangle \in RR \wedge \langle r, t \rangle \in RR \} \\
 \text{IN } TCR(S) &
 \end{aligned}$$

We now test that these four definitions are equivalent. Since it's unlikely that all four are wrong in the same way, their equivalence makes it highly probable that they're correct.

ASSUME  $\forall N \in 0 \dots 3 :$

$$\begin{aligned} \forall R \in \text{SUBSET } ((1 \dots N) \times (1 \dots N)) : & \wedge TC1(R) = TC2(R) \\ & \wedge TC2(R) = TC3(R) \\ & \wedge TC3(R) = TC4(R) \end{aligned}$$

Sometimes we want to represent a relation as a Boolean-valued operator, so we can write  $s R t$  as  $R(s, t)$ . This representation is less convenient for manipulating relations, since an operator is not an ordinary value the way a function is. For example, since TLA+ does not permit us to define operator-valued operators, we cannot define a transitive closure operator  $TC$  so  $TC(R)$  is the operator that represents the transitive closure. Moreover, an operator  $R$  by itself cannot represent a relation; we also have to know what set it is an operator on. (If  $R$  is a function, its domain tells us that.)

However, there may be situations in which you want to represent relations by operators. In that case, you can define an operator  $TC$  so that, if  $R$  is an operator representing a relation on  $S$ , and  $TCR$  is the operator representing its transitive closure, then

$$TCR(s, t) = TC(R, S, s, t)$$

for all  $s, t$ . Here is the definition. (This assumes that for an operator  $R$  on a set  $S$ ,  $R(s, t)$  equals FALSE for all  $s$  and  $t$  not in  $S$ .)

$$\begin{aligned} TC5(R(-, -), S, s, t) & \triangleq \\ \text{LET } CR[n \in Nat, v \in S] & \triangleq \\ \quad \text{IF } n = 0 \text{ THEN } R(s, v) & \\ \quad \text{ELSE } \vee CR[n - 1, v] & \\ \quad \vee \exists u \in S : CR[n - 1, u] \wedge R(u, v) & \\ \text{IN } \wedge s \in S & \\ \wedge t \in S & \\ \wedge CR[Cardinality(S) - 1, t] & \end{aligned}$$

Finally, the following assumption checks that our definition  $TC5$  agrees with our definition  $TC1$ .

$$\begin{aligned} \text{ASSUME } \forall N \in 0 \dots 3 : \forall R \in \text{SUBSET } ((1 \dots N) \times (1 \dots N)) : & \\ \quad \text{LET } RR(s, t) \triangleq \langle s, t \rangle \in R & \\ \quad S \triangleq \text{Support}(R) & \\ \quad \text{IN } \forall s, t \in S : & \\ \quad \quad TC5(RR, S, s, t) \equiv (\langle s, t \rangle \in TC1(R)) & \end{aligned}$$